DMAP Alters to Apply Modal Damping and Obtain Dynamic Loading Output for Superelements

Presented at the 1993 MSC World Users' Conference

by

Ted L. Rose

Staff Scientist, Dynamics Specialist

The MacNeal–Schwendler Corp.

Los Angeles, Ca.

Abstract:

A pair of DMAP alters are presented for use in the structured superelement dynamic solutions in MSC/NASTRAN Version 67.5. The first DMAP alter allows the entry of a "modal damping" matrix for superelements by using TABDMP1 entries and the frequencies of the calculated component modes. This allows the use of damping results obtained from dynamic testing of individual components in an assembly analysis.

The second alter implements the calculation and output (including plots and PSD) of the applied dynamic loads on superelements. This provides the user with a tool to verify that dynamic loads applied to superelements were entered properly.

Introduction

This paper provides two DMAP alters for MSC/NASTRAN, version 67.5 which allow the user to specify modal damping for the component modes of superelements and to obtain dynamic loading information for superelements. Both are provided as a part of the sssalter library which is delivered with version 67.5. This alter library is provided in it's own directory on the delivery media. The directory is /misc/sssalter. There are more than 20 commonly requested DMAP alters which have been developed by MSC employees either in response to client requests or in order to test prototype features. These alters have not gone through the rigorous testing and verification necessary to become a standard part of MSC/NASTRAN, but they do provide users with the ability to use analysis methods beyond those available in the program at the current time. The alters are provided along with "read–me" (.rdm) files. Additional documentation, mentioned in the "read–me" files, is available on request from your local MSC support office.

As anyone who has met me knows, I am an advocate of using superelements as a tool to assist in solving almost all problems. When using superelements in a dynamic analysis, the only time that modal damping or dynamic loading information may be specified is at the residual level (although it is possible to apply dynamic loads on superelements). It is not currently possible to apply modal damping to the component modes of a superelement, nor is it possible to print or plot the dynamic loadings which are applied to superelements. The two DMAP alters provided with this paper allow the use of modal damping on the component modes of superelements and allow for printing and plotting of dynamic loads as applied to superelements.

Modal Damping on Component Modes

Often, when different companies (or groups) are working on a project, each component is built and tested individually before the final assembly is created. When this is done, individual models of each component are created and updated based on test results. When a modal test is performed, often modal damping is measured for the "component modes". If it is available, it is desirable to enter this damping into the model, rather than using an approximate "assembly' damping when the components are assembled into the entire model. One method of accomplishing this was presented last year at the 1992 World Users' Conference[1]. The first DMAP alter (listed in Appendix 1) in this paper provides an alternative method to specify modal damping for the component modes of each superelement. Standard MSC/NASTRAN input is used to describe modal damping for the superelements (SDAMP command in Case Control and TABDMP1 in Bulk Data), only now a different modal damping set may be used for each superelement as opposed to the traditional use of one set only on the assembly modes.

The DMAP alter uses the specified modal damping table to calculate a modal damping matrix for each superelement based on the frequencies of its component modes. If desired, modal damping may still be requested at the residual level, but it should be noted that any modal damping specified at this level will be added onto the reduced damping from the upstream superelements (this is the standard method used). This will also occur at collector superelements. If there is damping coming from upstream superelements, it will be added to any modal damping specified for a collector superelement or the residual structure.

Applying modal damping at the superelement level will almost always result in a "coupled" damping matrix for the residual structure. This means that the uncoupled solution algorithm will not be used. If you want to use the uncoupled algorithm with this alter, then you must specify PARAM,NONCUP,−2, which forces the program to use the uncoupled algorithm, even if there are off−diagonal terms in the matrices. This will force the use of an "equivalent" system modal damping which results from combining the individual modal damping terms, but will ignore the coupling terms created when the matrices are assembled.

Limitations:

1) The DMAP alter does not check to verify that component modes have been calculated for the superelement. If component modes do not exist, the alter will result in UFM 3001.

2) As implemented, if one superelement uses modal damping, all superelements processed in the same run must also have a modal damping called in Case Control. This can be avoided by either of two methods;

a) Processing selected superelements in each run. For example, superelements which have modal damping could be processed in the first run with the DMAP alter included and superelements which do not have modal damping could be processed during a restart run without the DMAP alter,

or

b) Selecting an SDAMP set which refers to a TABDMP1 entry with 0.0 modal damping for the superelements which do not have modal damping.

Sample problem:

A cantilever beam model (shown below) was run using two approaches to test the alter. First it was run as a residual−only model with modal damping to establish a benchmark. Then the entire model (except grid point 1) was placed into a superelement and modal damping was applied on the superelement modes (which, for this case, will be the same as the residual structure modes) and the results from the two runs were compared. The input files are listed in Appendices 2 and 3.



The answers from the two runs were identical, indicating that the DMAP alter worked properly.

Superelement Dynamic Loads

The second DMAP alter implements data recovery (printing, plotting, etc) for applied dynamic loads on superelements. Although the ability to apply dynamic loads on superelements has been a feature in MSC/NASTRAN for a long time, the actual dynamic loading information is calculated when preparing to solve the residual structure. This means that the loadings on individual locations within a superelement are not available for output. When the second DMAP alter is included in a dynamic analysis run, MSC/NASTRAN will calculate the dynamic loadings for the individual superelements and allow output (printing, plotting, etc) of any dynamic loadings applied on superelements.

The DMAP alter requires the standard Case Control for the dynamic loading, plus a complete description of the dynamic loading for each superelement. That is, the correct DLOAD, TSTEP (or FREQ) entry must appear in the appropriate superelement subcase in order for the program to calculate the applied dynamic loadings.

The following example shows how to obtain dynamic loading information for a superelement. The same cantilever beam is divided into superelement 1 and the residual structure. Grid point 10 lies at the free end and grid point 1 is constrained in all six dof. The dynamic load is a decaying 5 hz sine wave starting with an initial magnitude of 2.0, applied in the negative y–direction at grid point 8, which is interior to superelement 1.

Two files are shown. The first is the conventional input, which applies the loading and gives the solution without an opportunity print or plot the input loading. The second sample is the same model with the DMAP alter added. This run will give the same results, but also allows printout and plots of the loading applied to the superelement.

Selected parts of the conventional input file for this run follow:

```
ID CBAR, TEST PROBLEM
SOL 112
TIME 10
CEND
TITLE = MODAL TRANSIENT FOR SUPERELEMENT MODEL
subtitle = load applied in a superelement
LABEL = conventional input
  LOADSET = 101          $ select superelement loadings
DISP(PLOT) = ALL
SUBCASE 1                $ superelement subcase
$
SUPER = 10
METHOD = 1               $ request cms
SPC = 1
$
SUBCASE 5                $ residual subcase
  TSTEP   = 5            $ time step
SDAMP=99                 $ modal damping
SPC = 1
DLOAD = 10               $ dynamic loading
METHOD = 1               $ eigenvalue solution
BEGIN BULK
$            Time step selection
$             SID     N(1)    DT(1)    NO(1)
TSTEP           5      900     .001        5
$
$            Dynamic loading description
$
$             SID    DAREA    DELAY    TYPE      T1       T2       F     PHASE
TLOAD2         10       10                       0.       5.       10.           +T12
$               C        B
+T12          -1.
FORCE,12,8,,2.,0.,-1.,0.
$
LSEQ,101,10,12
```

```
$
$ DEFINE SUPERELEMENT
$
SESET,10,6,THRU,10
$
$ additional input not shown
.
.
.
ENDDATA
```

The same file modified to allow output of the loading applied on the superelement follows:

```
ID CBAR, TEST PROBLEM
SOL 112
TIME 10
include 'seloadsa.v675' $ DAMP alter
CEND
TITLE = MODAL TRANSIENT FOR SUPERELEMENT MODEL
subtitle = load applied in a superelement
LABEL = conventional input
$
  LOADSET = 101          $ select superelement loadings
$
SUBCASE 1                $ superelement subcase
SET 99 = 10
DISP = 99
$
SUPER = 10
  TSTEP   = 5                       $ time step
  DLOAD   = 10                      $ dynamic loading
  SET 80 = 8
  OLOAD   = 80                      $ loading output for grid 8
METHOD = 1               $ request cms
$
SPC = 1
$
SUBCASE 5                $ residual subcase
  TSTEP    = 5           $ time step
SDAMP=99                 $ modal damping
SPC = 1
DLOAD = 10               $ dynamic loading
METHOD = 1               $ eigenvalue solution
DISP(PLOT) = ALL
$
```

```
OUTPUT(XYPLOT)
$
$ plot applied dynamic loading on superelement
$
SEPLOT 10
 XAXIS = YES
 YAXIS = YES
 XGRID = YES
 YGRID = YES
 XTITLE = TIME
 YTITLE =    y  L O A D  O N  G R I D  9 3
XYPLOT oload /8(T2)
BEGIN BULK
.
IDENTICAL TO PREVIOUS INPUT
.
.
ENDDATA
```

The following is the xy–plot generated by this run. It shows the applied loading on grid point 8 as a function of time. As expected, the plot represents the applied loading on the superelement grid point.



Plot of Applied Load on Grid Point 8

Limitations

This DMAP alter is not designed to work if DAREA entrics cxist in the input file. In fact it will result in a fatal message if DAREA of TF entries are used and thc points referenced on them are not exterior to all superelements.

Conclusions:

Two DMAP alters are presented in this paper which provide additional capabilities when using superelements in dynamic analysis. One alter simply allows the user to review the applied loads, while the other actually modifies the solution by allowing input of modal damping at the superelement level.

References

1. Flannigan, C., et. al., "Implementation of the Benfield–Hruda Modal Synthesis Method in MSC/NASTRAN", Proceedings, MSC 1992 World Users' Conference.

2) MSC/NASTRAN User's Manual, Mike Reymond, Editor, The MacNeal–Schwendler Corporation, MSR–39, August, 1991.

Appendix 1 – DMAP Alter for Superelement Modal Damping

```
$
$ FILE SEDAMPA.V675
$  DMAP alter to include modal damping for superelements
$ FOR V67.5
$
$  6/25/1992
$
$ USER INPUT –
$
$  1) STANDARD INPUT FOR CMS
$
$  2) CASE CONTROL COMMAND SDAMP IN SUPERELEMENT SUBCASE
$
$  3) TABDMP1 ENTRY FOR THE SUPERELEMENT MATCHING CASE CONTROL
$
$  4) PARAM,NONCUP,–2 IF UNCOUPLED ANALYSIS IS DESIRED AT THE
$     ASSEMBLY LEVEL. NOTE: THIS PARAMETER WILL DISABLE THE
$     COUPLED SOLUTION OPTION, RESULTING IN THE PROGRAM
$     IGNORING ANY OFF–DIAGONAL TERMS IN THE MATRICES.
$
$ ————————————————————————————————
COMPILE PHASE1 SOUIN=MSCSOU NOREF LIST
ALTER 2
TYPE PARM,,I,N,NEED,BEXISTS,NMODES $
$
$ Add modal damping if requested
$
ALTER 69
$
If (SEID<> 0) THEN $
  DPD    ,,GPLS,SILS,USET,,/
      GPLD,SILD,USETD,TFPOOL,DLT,PSDL,FRL,NLFT,
      TRL,EED,EQDYN/–1/S,N,LUSETD/0/0/0/0/0/
      V,N,DYNSTAT=0/V,Y,NEWDYN=1/123/S,N,NOUE $
  GKAM   USETD,PHIOZ,,CMLAMA,DIT,,,,CASES/
      MHHS,BHHS,KHHS,PHIDHS/NOUE/LMODES/LFREQ/
      99999./–1/–1/–1/0/S,N,FMODE/KDAMP $
  PARAML BHHS//'PRESENCE'////S,N,BEXISTS $
  IF (BEXISTS<>–1) THEN $ MODAL DAMPING
   PARAML   CASES//'DTI'/1/149//S,N,SDAMP $
   MESSAGE //' ' $
   MESSAGE //'****  Modal damping set'/sdamp/
        ' requested by user for superelement'/seid $
   MESSAGE //' ' $
   MESSAGE //' (Note: This may result in a non–diagonal da'/
```

```
        'mping matrix at the assembly level)' $
   MESSAGE //' ' $
   MESSAGE //'**** IN ORDER TO FORCE UNCOUPLED ASSEMBLY AN'/
        'ALYSIS, SET PARAM,NONCUP,-2 ***' $
   MESSAGE //'**** THIS WILL FORCE UNCOUPLED ANALYSIS '/
        'WHETHER A DIAGONAL MODAL DAMPING MATRIX '/
        'EXISTS OR NOT ****' $
   PARAML BHHS//'TRAILER'/1/S,N,NMODES $
   IF (NOQSET = NMODES) THEN $
    UMERGE1 USET,BHHS,,,/BHHA/'A'/'Q'/'T' $
   ELSE $
    NEED = NOQSET-NMODES $
    MATGEN ,/PARTNA/6/NOASET/NOTSET/NMODES/NEED $
    MERGE  ,,,,BHHS,PARTNA,/BHHA/-1 $
   ENDIF $
    ADD BAA,BHHA/BAAH $
    PURGEX /BAA,,,,/ALWAYS $
    EQUIVX BAAH/BAA/ALWAYS $
  ENDIF $ MODAL DAMPING
$
ENDIF $
$ ------------------------------------------------
COMPILE SEMR3, SOUIN=MSCSOU, LIST, NOREF
ALTER 102 $ REMOVE INREL IF SDAMPING IS USED
PARAML   CASES//'DTI'/1/149//S,N,SDAMP $
IF (SDAMP<=0) THEN $
ALTER 105 $
ELSE $
  MESSAGE //'**** DMAP INFORMATION MESSAGE ****' $
   MESSAGE //' ' $
   MESSAGE //' ' $
  MESSAGE //'**** USER HAS SPECIFIED THE USE OF MODAL DAMPING'/
        ' FOR THE CURRENT SUPERELEMENT' $
  MESSAGE //'**** THEREFORE THE INREL MODULE IS DISABLED FOR THE'/
        ' CURRENT SUPERELEMENT****' $
   MESSAGE //' ' $
  MESSAGE //'**** THIS MAY RESULT IN PROBLEMS IF FREE-FREE CMS HAS'/
        '  BEEN PERFORMED' $
   MESSAGE //' ' $
  MESSAGE //'**** IF 0.0 HZ MODES HAVE BEEN FOUND, THE INREL MODULE IS'/
        ' USED TO REMOVE POTENTIAL SINGULARITIES FROM THE MATRICES' $
   MESSAGE //' ' $
  MESSAGE //'**** USER ACTION - IF MODAL DAMPING IS TO BE USED ON A '/
        'SUPERELEMENT, DO NOT CALCULATE 0.0 HZ MODES FOR THAT'/
        ' SUPERELEMENT' $
  TYPE PARM,,I,N,NEED $
```

```
 PARAML PHIVZ//'TRAILER'/1/S,N,NMODES $
 IF (NOQSET = NMODES) THEN $
  EQUIVX PHIVZ/GOQ/ALWAYS $
 ELSE $
  NEED = NOQSET–NMODES $
  MATGEN ,/PARTNA/6/NOQSET/0/NMODES/NEED $
  MERGE  ,,,PHIVZ,,PARTNA,/GOQ/1 $
 ENDIF $
ENDIF $
$
$
compile gma , souin=mscsou, list, noref
alter 163
PARAML   CASES//'DTI'/1/149//S,N,SDAMP $
IF (SDAMP =0) THEN $
 MESSAGE //'**** DMAP WARNING MESSAGE****' $
  MESSAGE //' ' $
 MESSAGE //'**** MODAL DAMPING DMAP IS IN USE FOR SUPERELEMENTS'/
      ' AND NO REQUEST EXISTS FOR MODAL DAMPING FOR THE'/
      ' RESIDUAL STRUCTURE ***' $
 MESSAGE //'**** THIS WILL RESULT IN THE SUPERELEMENT MODAL '/
      'DAMPING BEING DOUBLED ****' $
  MESSAGE //' ' $
 MESSAGE //'**** THE FIX FOR THIS IS EITHER' $
  MESSAGE //' ' $
 MESSAGE //'**** 1) TO INCLUDE AN SDAMP COMMAND'/
      ' FOR THE RESIDUAL STRUCTURE POINTING TO A TABDMP1'/
      '     WITH DAMPING OF 0.0 FOR ALL FREQUENCIES' $
 MESSAGE //'     OR' $
 MESSAGE //'**** 2) INCLUDE A VALID SDAMP COMMAND FOR THE '/
      'RESIDUAL STRUCTURE SUBCASE' $
ENDIF $
$
$ END OF ALTER
$
```

Appendix 2 – Sample file 1 for Superelement Modal Damping

```
SOL 112
TIME 10
include 'sedampa.v675'
CEND
TITLE = MODAL TRANSIENT FOR SUPERELEMENT MODEL
subtitle = entire model in a superelement
LABEL = test of modal damping on superelement only
  LOADSET = 101
SPC = 1
SUBCASE 1
disp(plot) = all
  SDAMP    = 100
SUPER = 10
METHOD = 1
SUBCASE 5
  TSTEP    = 5
DLOAD = 10
METHOD = 1
BEGIN BULK
param,noncup,-2
SPC1,1,123456,1
TLOAD2       10      10                          0.      5.      10.         +T12
+T12        -1.
TSTEP         5     900     .001       5
FORCE,12,8,,2.,0.,0.,-1.
LSEQ,101,10,12
TABDMP1     100     CRIT                                                     +TDAMP
+TDAMP       0.      .01     200.      .01     ENDT
EIGR,1,MGIV,0.,100.,,10
SPOINT,1001,THRU,1010
SEQSET1,10,0,1001,THRU,1010
GRID,1,,0.,0.,0.
=,*(1),=,*(10.),==
=(8)
GRID,11,,0.,1.,0.,,123456
CBAR,1,1,1,2,11
=,*(1),=,*(1),*(1),==
=(7)
SESET,10,2,THRU,11
SESET,10,12
PBAR,1,1,1.,10.,10.,10.
PARAM,AUTOSPC,YES
MAT1,1,30.+6,,.3,.283
SPC1,1,123456,1
ENDDATA
```

Appendix 3 – Sample file 2 for Superelement Modal Damping

```
ID CBAR, TEST PROBLEM
SOL 112
TIME 10
CEND
TITLE = MODAL TRANSIENT FOR SUPERELEMENT MODEL
LABEL = no superelements
  LOADSET = 101
  SDAMP   = 100
SPC = 1
SUBCASE 5
  TSTEP   = 5
SPC = 1
LABEL = RES STR - SOLVE PROBLEM FOR SYSTEM MODES
DLOAD = 10
METHOD = 1
set 99 = 10
DISP = 99
BEGIN BULK
SPC1,1,123456,1
TLOAD2        10      10                      0.      5.      10.       +T12
+T12          -1.
TSTEP          5      900     .001     5
FORCE,12,8,,2.,0.,0.,-1.
LSEQ,101,10,12
TABDMP1       100     CRIT                                                +TDAMP
+TDAMP        0.      .01     200.     .01     ENDT
EIGR,1,MGIV,0.,100.,,10
GRID,1,,0.,0.,0.
=,*(1),=,*(10.),==
=(8)
GRID,11,,0.,1.,0.,,123456
GRID,12,,0.,1.,0.,,123456
CBAR,1,1,1,2,11
=,*(1),=,*(1),*(1),==
=(2)
CBAR,5,1,5,6,12
=,*(1),=,*(1),*(1),==
=(3)
PBAR,1,1,1.,10.,10.,10.
PARAM,AUTOSPC,YES
MAT1,1,30.+6,,.3,.283
SPC1,1,123456,1
ENDDATA
```

Appendix 4 – DMAP Alter for Superelement Loads

```
$
$ FILE SELOADSA.V675
$  DMAP alter to allow for output of dynamic loads on superelements
$ FOR V67.5
$
$  9/17/1992  – initial implementation – no DAREA entries allowed with this
$             DMAP alter – they point to the residual structure dof
$             and will result in a FATAL
$
$ USER INPUT –
$
$  1) LOADSET AND ASSOCIATED LSEQ ENTRIES TO DESCRIBE THE LOADING
$
$  2) CASE CONTROL COMMANDS DLOAD, FREQ (OR TSTEP) IN (OR ABOVE)
$    SUPERELEMENT SUBCASE WITH
$    ASSOCIATED DLOAD, RLOADi, TLOADi, FREQ, TSTEP
$    ENTRIES IN BULK DATA
$
$  3) OLOAD REQUEST FOR THE DESIRED SUPERELEMENT GRID POINTS
$
$ ————————————————————————————————————————
$
$ Add superelement dynamic load output
$
COMPILE SELR, SOUIN=MSCSOU, LIST, NOREF $
$
alter 18
type parm,nddl,i,n,peid $
peid=seid $
type db,mgg,dynamics,slt,cases,bgpdts,cstms,dit,est,mpts $
If (SEID<> 0) THEN $  use PJ to get physically applied loads only
 DPD     DYNAMICS  ,GPLS,SILS,USET,SLT,PJ/GPLD,SILD,USETD,
        ,DLT,PSDL,FRL,NLFT,TRL,EED,EQDYN/–1/S,N,LUSETD/
        0/0/0/0/0/0/1/123/S,N,NOUE $
 if( APP='TRANRESP')then $
$ Transient analysis
   TRLG   CASES,USETD,DLT,,BGPDTS,SILS,CSTMS,TRL,DIT,
        GM,GOA,,EST,MPTS,MGG/
        PPT,PST,PDT,PD,PH,TOL/
        S,N,NOSET/S,N,PDTEPD $
   type parm,,i,n,get109 $
   get109 = getsys(get109,109) $
   PUTSYS(1,109)                  $
   call dbstore ppt,,,,//0/seid/'DBALL'/0 $
   PUTSYS(get109,109)             $
```

```
 else if( APP='FREQRESP') then $
$ frequency response
   FRLG   CASES,USETD,DLT,FRL,GM,GOA,DIT,/
        PPF,PSF,PDF,FOL,PHF/
        SOLTYP/S,N,FREQY/S,N,APP $
   get109 = getsys(get109,109) $
   PUTSYS(1,109)                    $
   call dbstore ppf,,,,//0/seid/'DBALL'/0 $
   PUTSYS(get109,109)               $
 endif $
endif $
COMPILE SEDRCVR, SOUIN=MSCSOU, LIST, NOREF $
ALTER 43
type parm,,i,n,itsok $
If (SEID<> 0) THEN $  use PJ to get physically applied loads only
 if( APP='TRANRESP')then $
  call dbfetch /ppt,,,,/0/seid/0/0/s,itsok $
  SDR2    CASEDR,CSTMS,MPTS,DIT,EQEXINS,,ETT,OLBM,BGPDTS,
      PPT,,,EST ,XYCDBDR/
      OPPT1,,,,,/APP/S,N,NOSORT2/NOCOMPS $
 else if( APP='FREQRESP') then $
  call dbfetch /ppf,,,,/0/seid/0/0/s,itsok $
  SDR2    CASEDR,CSTMS,MPTS,DIT,EQEXINS,,ETT,OLBM,BGPDTS,
      PPF,,,EST ,XYCDBDR/
      OPPT1,,,,,/APP/S,N,NOSORT2/NOCOMPS $
 endif $
 SDR3    OPPT1,,,,,/OPG2TR,,,,,/        $
 OFP    OPG2TR// $
 XYTRAN  XYCDBDR,OPG2TR,,,,/XYPLTTT/APP/'PSET'/S,N,PFILE/S,N,CARDNO/
      S,N,NOXYP/1  $
 XYPLOT XYPLTTT// $
ENDIF $
$ ------------------------------------------------
alter 185
$ before RANDOM
if (SEID<>0) then
 purgex /opg2,,,,// $
 equivx opg2tr/opg2/always $
endif $
```